MATHEMATICS 201-BNK-05
Advanced Calculus
Martin Huard
Winter 2011

Vector Valued Functions with Maple

Vectors in are defined using < > as parenthesis.
For example, the vector $\vec{u} = (1, -2, 4)$ is defined by

u:= < 1, -2, 4 > ;

Note that if you use square parenthesis

A:= [4, 5, 6] ;

then Maple will see A as a point, and not a vector. To make it a vector, you can use the Vector command.

Vector(A);

Operations with vectors

When working with vector valued functions on Maple, the first thing to do is to load the **VectorCalculus** package, which contains a lot of the commands that we need.

with(VectorCalculus):
BasisFormat(false):

The second command, **BasisFormat(false)** is added so that vectors will be expressed in the form that we usually work with. It is optional.

Maple can handle most of the common operations with vectors. For example, suppose we have $\vec{u} = (1, -2, 4)$ and $\vec{v} = (3, 2, -2)$, then $\vec{u} + \vec{v}$ and $4\vec{u}$ are given by

u:= <1,-2,4>;
v:= <3,2,-2>;
u+v;
4*u;

For the magnitude of a vector, the command **Norm** in used. For example, $\|\vec{u}\|$ is given by

Norm(u,2);

Note: the 2 in the command means that we take a square root. Were we to put a four, that is **Norm(u,4)**, then we would be evaluating $\sqrt[4]{1^4 + (-2)^4 + 4^4}$, which does not correspond to the magnitude as usually defined. Ergo, just always use a 2.

The command for the scalar product is **DotProduct**, and for the cross product, **CrossProduct**.
For example, $\vec{u} \bullet \vec{v}$ and $\vec{u} \times \vec{v}$ are given by

DotProduct(u,v);
CrossProduct(u,v);

Calculus with Vector Valued Functions.

The same operators as with real functions will work.

For example, if $\vec{r}(t) = (\cos(t), \sin(t), t)$ then $\vec{r}'(t)$, $\vec{r}(\frac{\pi}{4})$, and $\int_0^{\frac{\pi}{4}} \vec{r}(t)\,dt$ are given by

```
r:=< cos(t), sin(t), t >;
diff(r,t);
simplify( subs(t=Pi/4, diff(r,t) ) );
int(r,t=0..Pi/4);
```

where the command **subs** is used to replace *t* with a value, and **simplify** to obtain the answer.

Plotting Vectors

We can represent vectors graphically using the **arrow** command from the *plots* package. The **arrow** command has the format **arrow(base, dir, options)** where **base** is the base of the arrow, that is the point where the arrow starts, and **dir** is the direction (as a vector).

For example, the vector $\vec{u} = (1, -2, 4)$, starting at the point (0,3,0) is graphed with:

```
with(plots):
arrow([0,3,0],<1,-2,4>, axes=normal, labels=[x,y,z]);
```

The options **axes=normal** and **labels=[x,y,z]** are added to see the coordinate axes. The angle of vision can be moved by clicking on the graph and moving your mouse (with the button still pressed down). Try it!

Plotting Vector Valued Functions in $\mathbb{R}^2$

To graph curves given by a vector function in $\mathbb{R}^2$ with Maple, we use the command **plot([x(t), y(t), *t*=*a*..*b*],options)**. For example, the curve $\vec{r}(t) = (t^2 - 1, t^3)$ is plotted with.

```
plot( [ t^2-1,t^3, t=-2..2 ]);
```

Plotting Vector Valued Functions in $\mathbb{R}^3$

For curves in $\mathbb{R}^3$, we use the command **spacecurve([x(t), y(t), z(t)], t=a...b, options)** from the *plots* package. For example, to plot the helix $\vec{r}(t) = (\cos t, \sin t, t)$, we have

```
spacecurve([cos(t),sin(t),t],t=-Pi..5*Pi, axes=normal);
```

You can rotate the picture with your mouse to obtain different views.

For more than one plot, start by entering each curve individually (using "**:**" at the end not to have the output) and then use the command **display** (from the **plots** package). For example, plotting the helix with its tangent line at $t = \frac{3\pi}{4}$.

```
curve:=spacecurve( [cos(t), sin(t), t], t=-Pi..5*Pi, axes=normal):
tangent:=spacecurve([ -sqrt(2)/2-t*sqrt(2), sqrt(2)/2-t*sqrt(2), 3*Pi/4+2*t], t=-1..1):
display( curve, tangent);
```

Or, the tangent vector to the helix at $t = \frac{\pi}{2}$,

```
curve:=spacecurve([cos(t),sin(t),t],t=-Pi..5*Pi, axes=normal):
tangentvect:=arrow(<0,1,Pi/2>, <-1,0,1>, color=blue):
display(curve,tangentvect);
```